

NAG Library Routine Document

G13FEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13FEF estimates the parameters of a univariate regression-GJR GARCH(p, q) process (see Glosten *et al.* (1993)).

2 Specification

```

SUBROUTINE G13FEF (DIST, YT, X, LDX, NUM, IP, IQ, NREG, MN, NPAR, THETA,      &
                  SE, SC, COVR, LDICOVR, HP, ET, HT, LGF, COPTS, MAXIT,      &
                  TOL, WORK, LWORK, IFAIL)
INTEGER          LDX, NUM, IP, IQ, NREG, MN, NPAR, LDICOVR, MAXIT, LWORK,  &
                IFAIL
REAL (KIND=nag_wp) YT(NUM), X(LDX,*), THETA(NPAR), SE(NPAR), SC(NPAR),    &
                COVR(LDICOVR,NPAR), HP, ET(NUM), HT(NUM), LGF, TOL,      &
                WORK(LWORK)
LOGICAL          COPTS(2)
CHARACTER(1)     DIST

```

3 Description

A univariate regression-GJR GARCH(p, q) process, with q coefficients α_i , for $i = 1, 2, \dots, q$, p coefficients β_i , for $i = 1, 2, \dots, p$, and k linear regression coefficients b_i , for $i = 1, 2, \dots, k$, can be represented by:

$$y_t = b_o + x_t^T b + \epsilon_t \quad (1)$$

$$h_t = \alpha_0 + \sum_{i=1}^q (\alpha_i + \gamma I_{t-i}) \epsilon_{t-i}^2 + \sum_{i=1}^p \beta_i h_{t-i}, \quad t = 1, 2, \dots, T \quad (2)$$

where $I_t = 1$, if $\epsilon_t < 0$, $I_t = 0$, if $\epsilon_t \geq 0$, and $\epsilon_t | \psi_{t-1} = N(0, h_t)$ or $\epsilon_t | \psi_{t-1} = S_t(df, h_t)$. Here S_t is a standardized Student's t -distribution with df degrees of freedom and variance h_t , T is the number of terms in the sequence, y_t denotes the endogenous variables, x_t the exogenous variables, b_o the regression mean, b the regression coefficients, ϵ_t the residuals, h_t is the conditional variance, and ψ_t the set of all information up to time t .

G13FEF provides an estimate for $\hat{\theta}$, the parameter vector $\theta = (b_o, b^T, \omega^T)$ where $b^T = (b_1, \dots, b_k)$, $\omega^T = (\alpha_0, \alpha_1, \dots, \alpha_q, \beta_1, \dots, \beta_p, \gamma)$ when $DIST = 'N'$ and $\omega^T = (\alpha_0, \alpha_1, \dots, \alpha_q, \beta_1, \dots, \beta_p, \gamma, df)$ when $DIST = 'T'$.

MN, NREG can be used to simplify the GARCH(p, q) expression in (1) as follows:

No Regression and No Mean

$$y_t = \epsilon_t,$$

$$MN = 0,$$

$$NREG = 0 \text{ and}$$

$$\theta \text{ is a } (p + q + 2) \text{ vector when } DIST = 'N', \text{ and a } (p + q + 3) \text{ vector when } DIST = 'T'.$$

No Regression

$$y_t = b_o + \epsilon_t,$$

$$MN = 1,$$

$$NREG = 0 \text{ and}$$

θ is a $(p + q + 3)$ vector when DIST = 'N', and a $(p + q + 4)$ vector when DIST = 'T'.

Note: if the $y_t = \mu + \epsilon_t$, where μ is known (not to be estimated by G13FEF) then (1) can be written as $y_t^\mu = \epsilon_t$, where $y_t^\mu = y_t - \mu$. This corresponds to the case **No Regression and No Mean**, with y_t replaced by $y_t - \mu$.

No Mean

$$y_t = x_t^T b + \epsilon_t,$$

$$MN = 0,$$

$$NREG = k \text{ and}$$

θ is a $(p + q + k + 2)$ vector when DIST = 'N', and a $(p + q + k + 3)$ vector when DIST = 'T'.

4 References

Bollerslev T (1986) Generalised autoregressive conditional heteroskedasticity *Journal of Econometrics* **31** 307–327

Engle R (1982) Autoregressive conditional heteroskedasticity with estimates of the variance of United Kingdom inflation *Econometrica* **50** 987–1008

Engle R and Ng V (1993) Measuring and testing the impact of news on volatility *Journal of Finance* **48** 1749–1777

Glosten L, Jagannathan R and Runkle D (1993) Relationship between the expected value and the volatility of nominal excess return on stocks *Journal of Finance* **48** 1779–1801

Hamilton J (1994) *Time Series Analysis* Princeton University Press

5 Parameters

- 1: DIST – CHARACTER(1) *Input*
On entry: the type of distribution to use for e_t .
 DIST = 'N'
 A Normal distribution is used.
 DIST = 'T'
 A Student's t -distribution is used.
Constraint: DIST = 'N' or 'T'.
- 2: YT(NUM) – REAL (KIND=nag_wp) array *Input*
On entry: the sequence of observations, y_t , for $t = 1, 2, \dots, T$.
- 3: X(LDX,*) – REAL (KIND=nag_wp) array *Input*
Note: the second dimension of the array X must be at least NREG.
On entry: row t of X must contain the time dependent exogenous vector x_t , where $x_t^T = (x_t^1, \dots, x_t^k)$, for $t = 1, 2, \dots, T$.

- 4: LDX – INTEGER *Input*
On entry: the first dimension of the array X as declared in the (sub)program from which G13FEF is called.
Constraint: $LDX \geq NUM$.
- 5: NUM – INTEGER *Input*
On entry: T , the number of terms in the sequence.
Constraints:
 $NUM \geq \max(IP, IQ);$
 $NUM \geq NREG + MN.$
- 6: IP – INTEGER *Input*
On entry: the number of coefficients, β_i , for $i = 1, 2, \dots, p$.
Constraint: $IP \geq 0$ (see also NPAR).
- 7: IQ – INTEGER *Input*
On entry: the number of coefficients, α_i , for $i = 1, 2, \dots, q$.
Constraint: $IQ \geq 1$ (see also NPAR).
- 8: NREG – INTEGER *Input*
On entry: k , the number of regression coefficients.
Constraint: $NREG \geq 0$ (see also NPAR).
- 9: MN – INTEGER *Input*
On entry: if $MN = 1$, the mean term b_0 will be included in the model.
Constraint: $MN = 0$ or 1 .
- 10: NPAR – INTEGER *Input*
On entry: the number of parameters to be included in the model.
 $NPAR = 2 + IQ + IP + MN + NREG$ when $DIST = 'N'$ and
 $NPAR = 3 + IQ + IP + MN + NREG$ when $DIST = 'T'$.
Constraint: $NPAR < 20$.
- 11: THETA(NPAR) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the initial parameter estimates for the vector θ .
The first element must contain the coefficient α_0 and the next IQ elements contain the coefficients α_i , for $i = 1, 2, \dots, q$.
The next IP elements must contain the coefficients β_j , for $j = 1, 2, \dots, p$.
The next element must contain the asymmetry parameter γ .
If $DIST = 'T'$, the next element contains df , the number of degrees of freedom of the Student's t -distribution.
If $MN = 1$, the next element must contain the mean term b_0 .
If $COPTS(2) = .FALSE.$, the remaining NREG elements are taken as initial estimates of the linear regression coefficients b_i , for $i = 1, 2, \dots, k$.
On exit: the estimated values $\hat{\theta}$ for the vector θ .

The first element contains the coefficient α_o , the next IQ elements contain the coefficients α_i , for $i = 1, 2, \dots, q$.

The next IP elements are the moving average coefficients β_j , for $j = 1, 2, \dots, p$.

The next element contains the estimate for the asymmetry parameter γ .

If DIST = 'T', the next element contains an estimate for df , the number of degrees of freedom of the Student's t -distribution.

If MN = 1, the next element contains an estimate for the mean term b_o .

The final NREG elements are the estimated linear regression coefficients b_i , for $i = 1, 2, \dots, k$.

- 12: SE(NPAR) – REAL (KIND=nag_wp) array Output

On exit: the standard errors for $\hat{\theta}$.

The first element contains the standard error for α_o and the next IQ elements contain the standard errors for α_i , for $i = 1, 2, \dots, q$.

The next IP elements are the standard errors for β_j , for $j = 1, 2, \dots, p$.

The next element contains the standard error for γ .

If DIST = 'T', the next element contains the standard error for df , the number of degrees of freedom of the Student's t -distribution.

If MN = 1, the next element contains the standard error for b_o .

The final NREG elements are the standard errors for b_j , for $j = 1, 2, \dots, k$.

- 13: SC(NPAR) – REAL (KIND=nag_wp) array Output

On exit: the scores for $\hat{\theta}$.

The first element contains the score for α_o , the next IQ elements contain the scores for α_i , for $i = 1, 2, \dots, q$.

The next IP elements are the score for β_j , for $j = 1, 2, \dots, p$.

The next element contains the score for γ .

If DIST = 'T', the next element contains the score for df , the number of degrees of freedom of the Student's t -distribution.

If MN = 1, the next element contains the score for b_o .

The final NREG elements are the scores for b_j , for $j = 1, 2, \dots, k$.

- 14: COVR(LDCOVR,NPAR) – REAL (KIND=nag_wp) array Output

On exit: the covariance matrix of the parameter estimates $\hat{\theta}$, that is the inverse of the Fisher Information Matrix.

- 15: LDCOVR – INTEGER Input

On entry: the first dimension of the array COVR as declared in the (sub)program from which G13FEF is called.

Constraint: LDCOVR \geq NPAR.

- 16: HP – REAL (KIND=nag_wp) Input/Output

On entry: if COPTS(2) = .FALSE., HP is the value to be used for the pre-observed conditional variance; otherwise HP is not referenced.

On exit: if COPTS(2) = .TRUE., HP is the estimated value of the pre-observed conditional variance.

- 17: ET(NUM) – REAL (KIND=nag_wp) array Output
On exit: the estimated residuals, ϵ_t , for $t = 1, 2, \dots, T$.
- 18: HT(NUM) – REAL (KIND=nag_wp) array Output
On exit: the estimated conditional variances, h_t , for $t = 1, 2, \dots, T$.
- 19: LGF – REAL (KIND=nag_wp) Output
On exit: the value of the log-likelihood function at $\hat{\theta}$.
- 20: COPTS(2) – LOGICAL array Input
On entry: the options to be used by G13FEF.
 COPTS(1) = .TRUE.
 Stationary conditions are enforced, otherwise they are not.
 COPTS(2) = .TRUE.
 The routine provides initial parameter estimates of the regression terms, otherwise these are to be provided by you.
- 21: MAXIT – INTEGER Input
On entry: the maximum number of iterations to be used by the optimization routine when estimating the GARCH(p, q) parameters. If MAXIT is set to 0, the standard errors, score vector and variance-covariance are calculated for the input value of θ in THETA; however the value of θ is not updated.
Constraint: MAXIT ≥ 0 .
- 22: TOL – REAL (KIND=nag_wp) Input
On entry: the tolerance to be used by the optimization routine when estimating the GARCH(p, q) parameters.
- 23: WORK(LWORK) – REAL (KIND=nag_wp) array Workspace
 24: LWORK – INTEGER Input
On entry: the dimension of the array WORK as declared in the (sub)program from which G13FEF is called.
Constraint: LWORK $\geq (\text{NREG} + 3) \times \text{NUM} + \text{NPAR} + 403$.
- 25: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G13FEF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NREG < 0,
 or MN > 1,
 or MN < 0,
 or IQ < 1,
 or IP < 0,
 or NPAR ≥ 20,
 or NPAR has an invalid value,
 or LDICOVR < NPAR,
 or LDX < NUM,
 or DIST ≠ 'N',
 or DIST ≠ 'T',
 or MAXIT < 0,
 or NUM < max(IP, IQ),
 or NUM < NREG + MN.

IFAIL = 2

On entry, LWORK < (NREG + 3) × NUM + NPAR + 403.

IFAIL = 3

The matrix X is not full rank.

IFAIL = 4

The information matrix is not positive definite.

IFAIL = 5

The maximum number of iterations has been reached.

IFAIL = 6

The log-likelihood cannot be optimized any further.

IFAIL = 7

No feasible model parameters could be found.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example fits a GARCH(1,1) model with Student's t -distributed residuals to some simulated data.

The process parameter estimates, $\hat{\theta}$, are obtained using G13FEF, and a four step ahead volatility estimate is computed using G13FFF.

The data was simulated using G05PFF.

9.1 Program Text

```

Program g13fefe

!      G13FEF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
Use nag_library, Only: g13fef, g13fff, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: gamma, hp, lgf, tol
Integer                    :: i, ifail, ip, iq, l, ldcovr, ldx,      &
                           lwork, maxit, mn, npar, nreg, nt,      &
                           num, pgamma
Logical                    :: tdist
Character (1)              :: dist
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: covr(:,,:), et(:), fht(:), ht(:),      &
                                   sc(:), se(:), theta(:), work(:),      &
                                   x(:,,:), yt(:)
Logical                    :: copts(2)
!      .. Executable Statements ..
Write (nout,*) 'G13FEF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Read in the problem size
Read (nin,*) num, mn, nreg

      ldx = num
      Allocate (yt(num),x(ldx,nreg))

!      Read in the series
Read (nin,*) yt(1:num)

!      Read in the exogenous variables
If (nreg>0) Then
      Read (nin,*)(x(i,1:nreg),i=1,num)
End If

!      Read in details of the model to fit
Read (nin,*) dist, ip, iq

!      Read in control parameters
Read (nin,*) copts(1:2), maxit, tol

!      Calculate NPAR
npar = 2 + iq + ip + mn + nreg
If (dist=='T' .Or. dist=='t') Then
      npar = npar + 1
      tdist = .True.
Else
      tdist = .False.
End If

      ldcovr = npar
      lwork = (nreg+3)*num + npar + 403
      Allocate (theta(npar),se(npar),sc(npar),covr(ldcovr,npar),et(num), &
               ht(num),work(lwork))

!      Read in initial values
!      alpha_0
Read (nin,*) theta(1)
l = 2

```

```

!   alpha_i
   If (iq>0) Then
     Read (nin,*) theta(1:(1+iq-1))
     l = l + iq
   End If
!   beta_i
   If (ip>0) Then
     Read (nin,*) theta(1:(1+ip-1))
     l = l + ip
   End If
!   gamma
   Read (nin,*) theta(l)
   pgamma = l
   l = l + 1
!   degrees of freedom
   If (tdist) Then
     Read (nin,*) theta(l)
     l = l + 1
   End If
!   mean
   If (mn==1) Then
     Read (nin,*) theta(l)
     l = l + 1
   End If
!   Regression parameters and pre-observed conditional variance
   If (.Not. copts(2)) Then
     Read (nin,*) theta(1:(1+nreg-1))
     Read (nin,*) hp
   End If

!   Fit the GARCH model
   ifail = -1
   Call g13fef(dist,yt,x,ldx,num,ip,iq,nreg,mn,npar,theta,se,sc,covr, &
     ldcovr,hp,et,ht,lgf,copts,maxit,tol,work,lwork,ifail)
   If (ifail/=0) Then
     If (ifail/=5 .And. ifail/=6) Then
       Go To 100
     End If
   End If

!   Read in forecast horizon
   Read (nin,*) nt

   Allocate (fht(nt))

!   Extract the estimate of the asymmetry parameter from theta
   gamma = theta(pgamma)

!   Calculate the volatility forecast
   ifail = 0
   Call g13fff(num,nt,ip,iq,theta,gamma,fht,ht,et,ifail)

!   Output the results
   Write (nout,*) '           Parameter           Standard'
   Write (nout,*) '           estimates           errors '
!   Output the coefficient alpha_0
   Write (nout,99999) 'Alpha', 0, theta(1), se(1)
   l = 2
!   Output the coefficients alpha_i
   If (iq>0) Then
     Write (nout,99999)('Alpha',i-1,theta(i),se(i),i=l,l+iq-1)
     l = l + iq
   End If
   Write (nout,*)
!   Output the coefficients beta_j
   If (ip>0) Then
     Write (nout,99999)('Beta',i-1+1,theta(i),se(i),i=l,l+ip-1)
     l = l + ip
     Write (nout,*)
   End If
!   Output the estimated asymmetry parameter, gamma

```



```

Write (nout,99998) ' Gamma', theta(1), se(1)
Write (nout,*)
l = l + 1
! Output the estimated degrees of freedom, df
If (dist=='T') Then
  Write (nout,99998) ' DF', theta(1), se(1)
  Write (nout,*)
  l = l + 1
End If
! Output the estimated mean term, b_0
If (mn==1) Then
  Write (nout,99999) ' B', 0, theta(1), se(1)
  l = l + 1
End If
! Output the estimated linear regression coefficients, b_i
If (nreg>0) Then
  Write (nout,99999)(' B',i-l+1,theta(i),se(i),i=l,l+nreg-1)
End If
Write (nout,*)

! Display the volatility forecast
Write (nout,*)
Write (nout,99997) 'Volatility forecast = ', fht(nt)
Write (nout,*)

100 Continue

99999 Format (1X,A,I0,1X,2F16.2)
99998 Format (1X,A,1X,2F16.2)
99997 Format (1X,A,F12.2)
End Program g13fefe

```

9.2 Program Data

```

G13FEF Example Program Data
100 1 2                                :: NUM,MN,NREG
7.23 6.75 7.21 7.08 6.60
6.59 7.00 7.06 6.82 6.99
7.05 6.12 7.47 6.99 7.26
6.42 7.12 6.77 7.32 6.03
6.78 7.04 6.27 7.30 7.71
6.62 8.13 7.69 7.62 6.64
8.16 6.95 7.15 7.61 7.42
7.56 8.25 7.43 7.84 7.24
7.63 8.45 8.17 7.40 7.62
8.89 8.14 8.90 7.79 7.19
7.55 7.41 7.93 7.43 8.87
7.27 8.09 7.15 8.21 8.19
7.84 7.99 8.90 8.24 7.97
8.30 8.23 7.98 7.73 8.50
7.71 7.70 8.61 7.68 8.66
8.85 8.09 7.45 6.15 6.28
7.59 6.78 9.32 9.16 8.77
8.27 7.24 7.73 9.01 9.09
7.55 8.64 7.97 8.20 7.72
8.47 8.06 5.55 8.75 10.15          :: End of Y
2.40 0.12          2.40 0.12
2.40 0.13          2.40 0.14
2.40 0.14          2.40 0.15
2.40 0.16          2.40 0.16
2.40 0.17          2.41 0.18
2.41 0.19          2.41 0.19
2.41 0.20          2.41 0.21
2.41 0.21          2.41 0.22
2.41 0.23          2.41 0.23
2.41 0.24          2.42 0.25
2.42 0.25          2.42 0.26
2.42 0.26          2.42 0.27
2.42 0.28          2.42 0.28
2.42 0.29          2.42 0.30

```

```

2.42 0.30      2.43 0.31
2.43 0.32      2.43 0.32
2.43 0.33      2.43 0.33
2.43 0.34      2.43 0.35
2.43 0.35      2.43 0.36
2.43 0.37      2.44 0.37
2.44 0.38      2.44 0.38
2.44 0.39      2.44 0.39
2.44 0.40      2.44 0.41
2.44 0.41      2.44 0.42
2.44 0.42      2.45 0.43
2.45 0.43      2.45 0.44
2.45 0.45      2.45 0.45
2.45 0.46      2.45 0.46
2.45 0.47      2.45 0.47
2.45 0.48      2.46 0.48
2.46 0.49      2.46 0.49
2.46 0.50      2.46 0.50
2.46 0.51      2.46 0.51
2.46 0.52      2.46 0.52
2.46 0.53      2.47 0.53
2.47 0.54      2.47 0.54
2.47 0.54      2.47 0.55
2.47 0.55      2.47 0.56
2.47 0.56      2.47 0.57
2.47 0.57      2.48 0.57
2.48 0.58      2.48 0.58
2.48 0.59      2.48 0.59
2.48 0.59      2.48 0.60
2.48 0.60      2.48 0.61
2.48 0.61      2.49 0.61
2.49 0.62      2.49 0.62
2.49 0.62      2.49 0.63
2.49 0.63      2.49 0.63
2.49 0.64      2.49 0.64
2.49 0.64      2.50 0.64
'T' 1 1
T T 200 0.00001
0.025
0.050
0.400
0.045
3.250
1.500
4
:: End of X
:: DIST,IP,IQ
:: COPTS,MAXIT,TOL
:: ALPHA_0
:: ALPHA_I
:: BETA_I
:: GAMMA
:: DF
:: MEAN
:: NT

```

9.3 Program Results

G13FEF Example Program Results

	Parameter estimates	Standard errors
Alpha0	0.08	0.12
Alpha1	0.00	0.85
Beta1	0.67	0.19
Gamma	0.35	0.63
DF	5.03	5.13
B0	50.22	3.33
B1	-18.48	1.43
B2	6.45	0.54

Volatility forecast = 0.61
